A maker of personal computers sells $15\,000$ units per month of a certain model. The cost of manufacture is $$600$/unit, and the wholesale price is $$850$. During the last quarter the maker lowered the price $$100$ in some markets, and it resulted in 40% increase in sales. The company has been advertising its product nationwide at a cost of $$4\,000$ per month. The marketing consultant claims that increasing the advertising budget by $$2\,000$/month would result in a sales increase of 400 units/month. Management has agreed to consider an increase in advertising budget to no more that $$10\,000$/month.

Graph level curves of the objective function for profit $P(x, y)$ where x is the wholesale price and y is the advertising budget for $x \in [600, 1200]$ and $y \in [1000, 12000]$. Overlay the advertising-budget constraint on the same plot.

```
using Symbolics
```

```
D (generic function with 1 method)
```

```
D(f,x)=expand_derivatives(Differential(x)(f))
```

```
[x, y]
```

```
@variables x,y
```

```
Attempting to print a symbolic expression in a Pluto notebook. Please run `import L
atexify` to enable pretty-printing of symbolic expressions. This warning will only
display once.
```

There are two natural ways to interpret how changing the price affects the quantity sold. The first way compounds the effects of the advertizing as

$$q(x, y) = \left(15000 + \frac{400}{2000}(y - 4000)\right)\left(1 + \frac{40}{10000}(850 - x)\right)$$

while the second way treats the increase in sales due to advertizing separate from the increase due to discounting the price as

$$q(x, y) = 15000\left(1 + \frac{40}{10000}(850 - x)\right) + \frac{400}{2000}(y - 4000).$$

Since first way leads to a quadratic equation when we later solve $\nabla P = \lambda \nabla g$ ignore that and continue with the second interpretation of q which leads to a linear system.

```
q (generic function with 1 method)
```

```
# This is the second option for q which is apparently the correct one
q(x,y)=15000*(1+40//10000*(850-x))+400//2000*(y-4000)
```

```
R (generic function with 1 method)
```

```
R(x,y)=x*q(x,y)
```

```
C (generic function with 1 method)
```

```
C(x,y)=600*q(x,y)+y
```

```
P (generic function with 1 method)
```

```
P(x,y)=R(x,y)-C(x,y)
```

```
# Using display causes the output to wrap
display(P(x,y))
```

```
-600(15000(1 + (1//250)*(850 - x)) + (1//5)*(-4000 + y)) - y + (15000(1 + (1//2
50)*(850 - x)) + (1//5)*(-4000 + y))*x
```

```
using Plots
```

```
g (generic function with 1 method)
```

```
g(x,y)=y
```




```
begin
    plt=contour(600:10:1200,1000:100:12000,g,levels=[10000]);
    contour!(plt,600:10:1200,1000:100:12000,P)
end

```

```
Multiple series with different levels share a colorbar.
Colorbar may not reflect all series correctly.
```

Determine the price x and the advertising budget y that will maximize profit subject to the constraint. Model as a constrained optimization problem and solve using the method of Lagrange multipliers.

What is $P(x, y)$ at the optimal values of x and y?

```
dPdx = (101200//1) - (120//1)*x + (1//5)*y
```

```
dPdx=expand(D(P(x,y),x))
```

```
dPdy = -(121//1) + (1//5)*x
```

```
dPdy=expand(D(P(x,y),y))
```

```
dgdx = 0
```

```
dgdx=expand(D(g(x,y),x))
```

```
dgdy = 1
```

```
dgdy=expand(D(g(x,y),y))
```

```
[lambda]
```

```
@variables lambda
```

```
S1 =  [(860//1), (10000//1), (51//1)]
```

```
S1=Symbolics.solve_for(
    [dPdx~lambda*dgdx,dPdy~lambda*dgdy,g(x,y)~10000],[x,y,lambda]
)
```

```
x0 = 860.0
```

```
x0=Float64(Symbolics.toexpr(S1[1]))
```

```
y0 = 10000.0
```

```
y0=Float64(Symbolics.toexpr(S1[2]))
```

```
lambda0 = 51.0
```

```
lambda0=Float64(Symbolics.toexpr(S1[3]))
```

```
# use formatted printing to display answer in dollars and cents
using Printf
```

The profit is maximixed when the wholesale price is $x =$ $860.00 and the advertising budget is $y =$ $10000.00. In this case the maximum profit is $4046000.00

Determine the sensitivity of the decision variables (price and advertising) to price elasticity (the 40% number). In particular, let α be the price elasticity and compute $S(x, \alpha)$ and $S(y, \alpha)$.

```
[alpha]
```

```
@variables alpha
```

In order to finish this modeling problem, I define new functions that are the same as those in the previous part, with an additional suffix and the parameter α.

In my answer to Quiz 2 I was able to write something like

```
alpha=40.0
```

to find the maximum and then write

```
@variables alpha
```

to dynamically change the meaning of alpha in the already defined functions and then find the sensitivity.

With a Pluto worksheet the interface does not allow one to redefine *anything* in the middle of the worksheet. This allows the notebook interface to track data dependencies and update the cells automatically based on the dependency relations between them.

```
qa (generic function with 1 method)
```

```
qa(x,y)=15000*(1+alpha*1//10000*(850-x))+400//2000*(y-4000)
```

```
Ra (generic function with 1 method)
```

```
Ra(x,y)=x*qa(x,y)
```

```
Ca (generic function with 1 method)
```

```
Ca(x,y)=600*qa(x,y)+y
```

```
Pa (generic function with 1 method)
```

```
Pa(x,y)=Ra(x,y)-Ca(x,y)
```

```
dPadx = (14200//1) + (2175//1)*alpha + (1//5)*y - (3//1)*alpha*x
```

```
dPadx=expand(D(Pa(x,y),x))
```

```
dPady = -(121//1) + (1//5)*x
```

```
dPady=expand(D(P(x,y),y))
```

Although it is possible to solve for x and y we use implicit differentiation here to find the derivatives without needing to explicity solve for x and y. Write

$$u = \partial P/\partial x - \lambda \partial g/\partial x, \qquad v = \partial P/\partial x - \lambda \partial g/\partial x \qquad \text{and} \qquad w = g(x,y) - 10000$$

so differentiating with respect to α implies

$$\frac{du}{d\alpha} = \frac{\partial u}{\partial \alpha} + \frac{\partial u}{\partial x}\frac{dx}{d\alpha} + \frac{\partial u}{\partial y}\frac{dy}{d\alpha} + \frac{\partial u}{\partial \lambda}\frac{d\lambda}{d\alpha} = 0$$

$$\frac{dv}{da} = \frac{\partial v}{\partial \alpha} + \frac{\partial v}{\partial x}\frac{dx}{d\alpha} + \frac{\partial v}{\partial y}\frac{dy}{d\alpha} + \frac{\partial v}{\partial \lambda}\frac{d\lambda}{d\alpha} = 0$$

and

$$\frac{dw}{da} = \frac{\partial w}{\partial \alpha} + \frac{\partial w}{\partial x}\frac{dx}{d\alpha} + \frac{\partial w}{\partial y}\frac{dy}{d\alpha} + \frac{\partial w}{\partial \lambda}\frac{d\lambda}{d\alpha} = 0.$$

No matter how complicated the functional relationships for x, y and λ actually are, implicit differentiation always yields linear equations in the derivatives.

Now solve these three linear equations in the unknowns $dx/d\alpha$, $dy/d\alpha$ and $d\lambda/d\alpha$.

```
u = (14200//1) + (2175//1)*alpha + (1//5)*y - (3//1)*alpha*x
```

```
u=dPadx-lambda*dgdx
```

```
v = -(121//1) - lambda + (1//5)*x
```

```
v=dPady-lambda*dgdy
```

```
w = -10000 + y
```

```
w=g(x,y)-10000
```

```
A = 3×3 Matrix{Num}:
 (-3//1)*alpha  1//5   0
          1//5     0  -1
             0     1   0
```

```
A=[D(u,x) D(u,y) D(u,lambda);
   D(v,x) D(v,y) D(v,lambda);
   D(w,x) D(w,y) D(w,lambda)]
```

```
b =  [-(2175//1) + (3//1)*x, 0, 0]
```

```
b=-[D(u,alpha),D(v,alpha),D(w,alpha)]
```

```
X =  [((2175//1) - (3//1)*x) / ((3//1)*alpha), 0//1, ((2175//1) - (3//1)*x) / (15alpha)]
```

```
X=A\b
```

At this point $X = [dx/d\alpha, dy/d\alpha, d\lambda/d\alpha]$. Now compute the sensitivities.

```
s0 =  [alpha ⇒ 40.0, x ⇒ 860.0, y ⇒ 10000.0, lambda ⇒ 51.0]
```

```
s0=[alpha=>40,x=>x0,y=>y0,lambda=>lambda0]
```

```
Sxa = -0.1569767441860465
```

```
Sxa=substitute(alpha/x*X[1],s0)
```

```
Sya = 0
```

```
Sya=substitute(alpha/y*X[2],s0)
```

It follows that $S(x, \alpha) =$ -0.1569767441860465 and $S(y, \alpha) =$ 0.